

ИМИТАЦИОННОЕ МОДЕЛИРОВАНИЕ

проф. Черненко В.М.

доц. Михеев А.В.

доц. Черненко М.В.

Учебно-методическое издание «Имитационное моделирование» представляет собой методические указания к лабораторным работам по курсу «Имитационное моделирование». Целью лабораторных работ является изучение построения имитационных моделей в среде GPSS. Изучаются действия операторов и описателей языка. Изучаются методы и особенности структур моделей. Выполняются практические работы по применению навыков построения моделей, анализа результатов моделирования и проведения математических экспериментов. Все работы должны быть выполнены на занятиях в лабораторном зале. В конце занятия студенты должны продемонстрировать преподавателю рабочую модель и все основные результаты в соответствии с заданием.

Задачи, решаемые в лабораторных работах

В процессе выполнения лабораторных работ студенты индивидуально должны выполнить поставленные задачи.

Основные требования к лабораторной работе:

- открыть в GPSS World модель;
- изучить алгоритм модели;
- провести несколько вариантов моделирования с изменением параметров модели;
- отобразить графически результаты моделирования.

Все действия студенты выполняют в интерактивном режиме в среде GPSS World и/или редакторе gpss-ee. После выполнения задания студенты демонстрируют работу модели, полученные отчеты, графические результаты и защищают работу, отвечая на контрольные вопросы преподавателя. Отчет подготавливается в текстовом процессоре (типа Word) с включением необходимых иллюстраций.

Основные требования

При разработке моделей важную роль играет отладчик, который встроен в среду моделирования (Blocks window). В режиме отладки можно проверить работоспособность модели и выполнить поиск ошибок самого разного характера. Отладчик позволяет проследить ход (по шагам) выполнения модели и одновременно получить текущие значения всех переменных (Expression window) и объектов модели (Facility window, Storage window), что позволяет установить моменты времени и операторы, в которых происходят коллизии, логические ошибки, закликивания и предпринять меры их устранения.

Все работы должны быть выполнены студентом на занятиях в лабораторном зале. В конце занятия студенты должны продемонстрировать рабочую модель, содержащую все основные результаты.

Для выполнения лабораторных работ потребуется использование следующих учебных и методических материалов:

- конспекты лекций по курсу «Имитационное моделирование»;
- методическое пособие по «Моделирование в среде GPSS World»;
- электронный справочник по языку GPSS.

Темы и задания лабораторных работ

Задания сгруппированы по сходным тематикам. На каждое задание отводится определенное время, в которое студент должен выполнить работу. Также ограничено общее время проведения всех заданий по одной тематике.

Лабораторная работа «Функции, многоканальные обработчики»

(Общее время выполнения — 4 часа)

Задание 1

(Время выполнения — 20 мин)

Изучите пример модели SCHR4A.GPS, где приведена типовая модель обслуживания потока транзактов.

```
Mean FUNCTION      Q$Wait,D4                      ;Мат.ожидание времени обработки
0,330/2,300/5,270/6,240
*
GENERATE      (Exponential(1,0,300))                ;Приход клиентов
QUEUE         Wait                                  ;Вход в очередь
SEIZE         Survu                                 ;Начало обслуживания в сервере
DEPART        Wait                                  ;Выход из очереди
ADVANCE       FN$Mean, (Exponential(1,0,1))          ;Время обслуживания
RELEASE       Survu                                 ;Освобождение сервера
TERMINATE     1
```

В этом примере среднее время обслуживания устройством Survu зависит от длины очереди ожидания обслуживания (например, ситуация обслуживания в буфете). Эта зависимость в модели задана функцией MEAN. Необходимо построить график функции MEAN и график изменения длины очереди WAIT.

Задание 2

(Время выполнения — 25 мин)

Изучите пример модели SCHR4B.GPS, где приведена модель магазина самообслуживания.

```
Carts      STORAGE      1000
Ayl1       FUNCTION      RN1,C2                      ;количество покупок в отделе 1
```

```

0,2/1,5
Ayl2 FUNCTION RN1,C2 ;количество покупок в отделе 2
0,3/1,6
Ayl3 FUNCTION RN1,C2 ;количество покупок в отделе 3
0,4/1,7
Cotym FUNCTION P1,C2 ;время обслуживания у кассира
0,3/18,54
Impul FUNCTION RN1,C2 ;количество покупок перед кассой
0,1/1,4
* модельный сегмент 1
GENERATE (Exponential(1,0,75)) ;Shoppers arrive
ENTER Carts ;Take a cart
TRANSFER .25,,Try2 ;Want to shop aisle 1?
ADVANCE 120,60 ;Time in aisle 1
ASSIGN 1, FN$Ayl1 ;Set P1 = Total items selected
Try2 TRANSFER .45,,Try3 ;Want to shop aisle 2?
ADVANCE 150,30 ;Time in aisle 2
ASSIGN 1+, FN$Ayl2 ;Set P1 = Total items selected
Try3 TRANSFER .18,,Out ;Want to shop aisle 3?
ADVANCE 120,45 ;Time in aisle 3
ASSIGN 1+, FN$Ayl3 ;Set P1 = Total items selected
Out QUEUE Checker ;Queue for checkout
ASSIGN 1+, FN$Impul ;Add to P1 items taken on impulse
SEIZE Checker ;Capture the checker
DEPART Checker ;Leave the checkout queue
ADVANCE FN$Cotym ;Checkout time
RELEASE Checker ;Free the checker
LEAVE Carts ;Return the cart
TERMINATE ;Leave the store
* модельный сегмент 2
GENERATE 28800 ;Timer arrives at end of 8-hours
TERMINATE 1 ;Shut off run

```

В магазине 3 отдела (aisle). С некоторой вероятностью покупатели делают покупки в некоторых отделах. Типовое количество покупок для каждого отдела задано через функции (Ayl1, Ayl2, Ayl3, Impul) и накапливается у транзактов в параметре локальной среды P1.

Время обслуживания в кассе (устройство checker) зависит от количества покупок и задано функцией Cotym. Необходимо построить график количества покупок на одного покупателя за 8 часов модельного времени.

Задание 3

(Время выполнения — 25 мин)

Изучите пример модели ORDERPNT.GPS, где приведена типовая модель работы склада.

```

INITIAL X$EOQ,500 ;Economic order qty.
INITIAL X$Point,600 ;Order point
INITIAL X$Stock,700 ;Set initial stock=700
Inventory TABLE X$Stock,0,50,20 ;Table of stock levels
Sales TABLE P$Demand,38,2,20 ;Table of sales levels
Var2 VARIABLE RN1@24+40
* модельный сегмент 1
GENERATE ,,,1
Again TEST L X$Stock,X$Point ;Order placed on successful test
ADVANCE 5 ;Lead time = 1 week
SAVEVALUE Stock+,X$EOQ ;Economic order

```

```

TRANSFER      ,Again                ;Cycle transaction again
*
  модельный сегмент 2
    GENERATE    1                    ;Daily demand xact
    ASSIGN      Demand,V$Var2        ;Assign daily demand
    TABULATE    Inventory             ;Record inventory
    TEST GE     X$Stock,P$Demand      ;Make sure order can be filled
    SAVEVALUE   Stock-,P$Demand      ;Remove demand from stock
    SAVEVALUE   Sold,P$Demand         ;X$Sold=Daily demand
    TABULATE    Sales                 ;Record daily sales
    TERMINATE   1                    ;Daily timer

```

Необходимо построить график изменения количества товара на складе за 90 дней модельного времени.

Задание 4

(Время выполнения — 30 мин)

Изучите пример модели MANUFACT.GPS, где приведена модель производственного участка со складом.

```

Sizeorder FUNCTION RN1,D7                ;Order size
.10,6/.35,12/.65,18/.80,24/.92,30/.97,36/1.0,48
Transit TABLE M1,.015,.015,20          ;Transit time
Number TABLE X1,100,100,20             ;No. packed each day
Ptime VARIABLE .0028#P1+0.0334          ;Packing time
Amount EQU 1000                          ;Initial stock amount
Stock STORAGE 4000                      ;Warehouse holds 4000 units
*
  модельный сегмент 1
    GENERATE (Exponential(1,0,0.25)) ;Order arrives
    ASSIGN 1,1,Sizeorder                ;P1=order size
    TEST GE S$Stock,P1,Stockout         ;Is stock sufficient?
    LEAVE Stock,P1                      ;Remove P1 from stock
    QUEUE Packing                       ;
    SEIZE Machine                        ;Get a machine
    DEPART Packing                       ;
    ADVANCE V$Ptime                      ;Packing time
    RELEASE Machine                     ;Free the machine
    SAVEVALUE 1+,P1                     ;Accumulate no. packed
    TABULATE Transit                    ;Record transit time
    TERMINATE
Stockout TERMINATE
*
  модельный сегмент 2
    GENERATE 0.75,0.08334,1             ;Xact every 40+/-5 mins
    ENTER Stock,60                      ;Make 60, Stock
                                         increased by 60
*
    TERMINATE
*
  модельный сегмент 3
    GENERATE 8                          ;Xact every day
    TABULATE Number
    SAVEVALUE 1,0
    TERMINATE 1
*
  модельный сегмент 4
    GENERATE ,,,1,10                    ;Initial stock xact
    ENTER Stock,Amount                  ;Set initial stock
    TERMINATE

```

Необходимо построить график изменения количества товара на складе и график времени изготовления деталей за 40 часов модельного времени.

Задание 5

(Время выполнения — 30 мин)

Изучите пример модели STOCKCTL.GPS, где приведена модель функционирования складов дистрибутора.

```

INITIAL      X1,3400           ;Fact warehouse inventory
INITIAL      X2,2100           ;Fact warehouse order pnt
INITIAL      X3,2300           ;Fact warehouse order qty
INITIAL      X$Stock1,430      ;Dist 1 stock initial
INITIAL      X$Stock2,600      ;Dist 2 stock initial
INITIAL      X$Stock3,1000     ;Dist 3 stock initial
INITIAL      X$EOQ1,115        ;Economic order qty 1
INITIAL      X$EOQ2,165        ;Economic order qty 2
INITIAL      X$EOQ3,200        ;Economic order qty 3
INITIAL      X$Point1,240      ;Order point 1
INITIAL      X$Point2,430      ;Order point 2
INITIAL      X$Point3,630      ;Order point 3
Demand1 VARIABLE (Normal(2,64,24))
Demand2 VARIABLE (Normal(3,128,32))
Demand3 VARIABLE (Normal(4,192,48))
Total        VARIABLE P1+P2+P3
Sales        TABLE          X5,200,200,20
Region_1     TABLE          X$Stock1,0,40,20
Region_2     TABLE          X$Stock2,0,40,20
Region_3     TABLE          X$Stock3,0,40,20
Factory      TABLE          X1,0,200,20
*            модельный сегмент 1
            GENERATE      ,,,1,2           ;Order point xact
Backhere     TEST LE        X1,X2           ;Factory order point?
            ADVANCE        4               ;Lead time is 4 weeks
            SAVEVALUE      1+,X3           ;Inv increase by order qty
            TRANSFER       ,Backhere       ;Cycle xact around
*            модельный сегмент 2
            GENERATE      1,,,1           ;First distributor
Distr1       TEST L        X$Stock1,X$Point1 ;Order point reached?
            ADVANCE        1               ;Lead time = 1 week
            SAVEVALUE      1-,X$EOQ1       ;Warehouse supplies
            SAVEVALUE      Stock1+,X$EOQ1   ;Distr invent increased
            TRANSFER       ,Distr1         ;Xact finished
*            модельный сегмент 3
            GENERATE      1,,,1           ;Second distributor
Distr2       TEST L        X$Stock2,X$Point2 ;Order point reached
            ADVANCE        1               ;Lead time = 1 week
            SAVEVALUE      1-,X$EOQ2       ;Warehouse supplies
            SAVEVALUE      Stock2+,X$EOQ2   ;Inventory increased
            TRANSFER       ,Distr2         ;Cycle xact around
*            модельный сегмент 4
            GENERATE      1,,,1           ;Third distributor
Distr3       TEST L        X$Stock3,X$Point3 ;Order point reached?
            ADVANCE        1               ;Lead time = 1 week
            SAVEVALUE      1-,X$EOQ3       ;Warehouse supplies EOQ
            SAVEVALUE      Stock3+,X$EOQ3   ;Distr invent increased
            TRANSFER       ,Distr3         ;Cycle xact around
*            модельный сегмент 5
            GENERATE      1,,,,3          ;Priority weekly demand
            ASSIGN         1,V$Demand1     ;P1 = Demand distr one
            ASSIGN         2,V$Demand2     ;P2 = Demand distr two
            ASSIGN         3,V$Demand3     ;P3 = Demand distr three
            SAVEVALUE      Stock1-,P1      ;Distr 1 Weeklly demand
            SAVEVALUE      Stock2-,P2      ;Distr 2 Weekly demand

```

```

SAVEVALUE Stock3-,P3           ;Distr 3 Weekly demand
SAVEVALUE 5+,V$Total           ;Accumulate total demand
TABULATE Region_1              ;Record invent distr 1
TABULATE Region_2              ;Record invent distr 2
TABULATE Region_3              ;Record invent distr 3
TABULATE Factory               ;Factory warehouse invent
TERMINATE 1
* модельный сегмент 6
GENERATE 4,,,1                 ;Low priority xact monthly
TABULATE Sales
SAVEVALUE 5,0                  ;Reset sales=0 each month
TERMINATE                       ;Xact finished

```

Необходимо построить график изменения количества товара на складах за 50 недель модельного времени.

Задание 6

(Время выполнения — 45 мин)

Постановка задачи самостоятельной работы

Сделать модель обслуживания заказов клиентов дистрибутора в его филиалах (с применением объектов типа storage) по данным складского процесса из задания 5. Построить графики изменения складских запасов для всех 4-х складов за 90 дней модельного времени.

Лабораторная работа «Логические ключи, порождение процессов»
(Общее время выполнения — 4 часа)

Задание 1

(Время выполнения — 30 мин)

Изучите пример модели LOCKSIMN.GPS, где приведена модель работы шлюза.

```
Upbarge  FUNCTION  X$Upcount,D6
1,.967/2,.767/3,.767/4,.767/5,.767/6,.767
Downbarge FUNCTION  X$Downcount,D6
1,.967/2,.767/3,.767/4,.767/5,.767/6,.767
Upq      QTABLE    Upq,.25,.25,20
Downq    QTABLE    Dnq,.25,.25,20
Upcount  TABLE    X$Upcount,2,2,20
Dncount  TABLE    X$Downcount,2,2,20
          INITIAL   X$Uplimit,6      ;No. of barges to go up
          INITIAL   X$Downlimit,6    ;No. of barges to go down
*        модельный сегмент 1
          GENERATE  1.67,.5,.67      ;Up barge arrives
          QUEUE     UPQ               ;Join queue
          GATE LR   Lock              ;Gate for the lock
          SEIZE     Lock              ;Get the lock
          SAVEVALUE Upcount+,1        ;Accumulate up number
          DEPART    Upq               ;Depart the queue
          ADVANCE   FN$Upbarge        ;Time to service barge
          TEST GE   X$Uplimit,X$Upcount,Swh1 ;Have enough passed?
          TEST NE   Q$Upq,0,Swh1      ;Check if Upq is zero
          RELEASE   Lock              ;Free the lock
          TERMINATE
*        ветвь переключения шлюза 1
Swh1     LOGIC S   Lock              ;Set lock the other way
          RELEASE   Lock              ;Free the lock
          TABULATE  Upcount           ;Record no. passed up
          SAVEVALUE Upcount,0        ;Set count to zero
          TERMINATE
*        модельный сегмент 2
          GENERATE  1.67,.5,1         ;Arrival of down barge
          QUEUE     Dnq               ;Enter queue
          GATE LS   Lock              ;Is lock set?
          SEIZE     Lock              ;Get the lock
          SAVEVALUE Downcount+,1      ;Accumulate down count
          DEPART    Dnq               ;Depart the queue
          ADVANCE   FN$Downbarge      ;Time for down barge
          TEST GE   X$Downlimit,X$Downcount,Swh2 ;Down count reached?
          TEST NE   Q$Dnq,0,Swh2     ;Any down barges left?
          RELEASE   Lock              ;Free the lock
          TERMINATE
*        ветвь переключения шлюза 2
Swh2     LOGIC R   Lock              ;Set lock for other way
          RELEASE   Lock              ;Free the lock
          TABULATE  Dncount           ;Record down count
          SAVEVALUE Downcount,0      ;Reset down count to zero
          TERMINATE
*        модельный сегмент 3
          GENERATE  24                 ;One xact per day
          TERMINATE 1                 ;One day has passed
```

Шлюз соединяет два уровня речного канала. Баржи подходят с двух сторон — сверху и снизу. Емкость шлюза ограничена шестью баржами. Время входа в

шлюз первой баржи из очереди больше, чем для последующих из-за учета времени освобождения шлюза.

Направление движения барж определяется логическим ключом с названием LOCK. Так, если он выключен (состояние reset), означает что происходит движение вверх по течению, если включен (состояние set) — то вниз по течению.

Необходимо промоделировать работу шлюза в течение месяца (30 дней).

Получите и проанализируйте таблицы распределения количества барж и времен ожидания в очереди для обоих направлений.

Задание 2

(Время выполнения — 30 мин)

Изучите пример модели SCHR6C.GPS, где представлена модель станка с периодической заменой запчастей.

```
*      модельный сегмент 1
      GENERATE      , , , 1      ;First worker arrives
Again SEIZE      Mac      ;Turn the machine on
      ADVANCE      (Normal(1,3500,700)) ;Part's lifetime elapses
      RELEASE      Mac      ;Turn the machine off
      ADVANCE      40      ;Remove the failed part
      SPLIT      1,Fetch      ;Send co-worker to fetch a spare
      SEIZE      Fixer      ;Capture the repairman
      ADVANCE      (Normal(1,80,5)) ;Repair proceeds
      RELEASE      Fixer      ;Free the repairman
      SAVEVALUE 1+,1      ;Update the number of good spares
      TERMINATE      ;Leave, co-worker will carry on
Fetch TEST G      X1,0      ;Wait (if necessary) for good one
      SAVEVALUE 1-,1      ;Update the number of good spares
      ADVANCE      60      ;Install the part
      TRANSFER      ,Again      ;Go turn on the machine
*      модельный сегмент 2
      GENERATE      (Exponential(1,0,90)) , , , , 1 ;'Other items' arrive at shop
      ADVANCE      ;Dummy advance block
      SEIZE      Fixer      ;Capture the repairman
      ADVANCE      80,40      ;Repair proceeds
      RELEASE      Fixer      ;Free the repairman
      TERMINATE      ;'Other item' leaves
*      модельный сегмент 3
      GENERATE      104000      ;Timer comes after 260 weeks
      TERMINATE 1      ;Shut off the run
```

В этой задаче моделируется работа сверлильного станка. На сверлильном станке сверло периодически выходит из строя (тупится). Токарь-станочник меняет его на запасное, если оно есть, и передает слесарю-инструментальщику для восстановления (заточки). Слесарь в основном занят изготовлением техоснастки и выполняет заточку сверла в свободное от основной работы время.

Необходимо промоделировать работу станка в течение года. Найти загрузку слесаря и станка, количество замен сверла. Построить график ожидания сверлом операции по заточке.

Задание 3

(Время выполнения — 30 мин)

Изучите пример модели SCHR5D.GPS, где приведена модель автозаправки.

```
Iat      FUNCTION      RN1,C7              ;Inter-arrival time distribution
0,0/.25,100/.48,200/.69,300/.81,400/.9,500/1,600
Stime    FUNCTION      RN1,C7              ;Service time distribution
0,100/.06,200/.21,300/.48,400/.77,500/.93,600/1,700
Store1   STORAGE       1                  ;Configure for 1-attendant case
Net      VARIABLE      SC$store1-75-30#R$Store1
*        модельный сегмент 1
          GENERATE      FN$Iat,,,,1        ;Cars approach station
          GATE LR        Lock                ;Station open
          ASSIGN         1,FN$Stime         ;Set P1 = latent service time
          TEST LE        Q1,Store1,Bybye    ;No. waiting <= no. in service?
Goin      QUEUE         1                  ;Yes, get in line
          ENTER          Store1             ;Capture and attendant
          DEPART         1                  ;Get out of line
          PRIORITY       2                  ;Raise priority to resolve
          ADVANCE        P1                 ;Servicing proceeds
Done      LEAVE         Store1             ;Free the attendant
Bybye     TERMINATE     ;Leave the station
*        модельный сегмент 2
          GENERATE      43200              ;Closeup time, owner arrives
          LOGIC S        Lock                ;Set 'not open' signal
          TEST E         N$Goin,N$Done      ;Wait until last car is serviced
          SAVEVALUE      1,V$Net           ;Record day's net profit
          TERMINATE     1                  ;Go home
```

На АЗС приезжают автомобили в интервале от 0 до 600 сек, что задано функцией IAT.

Длительность заправки от 100 до 700 с. Размер очереди не может становиться больше количества заправочных колонок. Через 12 ч приезжает инкассация, обслуживание клиентов прекращается, дообслуживаются все приехавшие, и фиксируется финансовый результат.

Необходимо провести моделирование при количестве колонок, меняющемся от 1 до 5, сравнить результаты моделирования.

Задание 4

(Время выполнения — 30 мин)

Изучите пример модели TVREPAIR.GPS, где приведена модель ремонтной мастерской.

```
*        модельный сегмент 1
          GENERATE      2400,480,,,1        ;Overhaul of a rented set
          QUEUE         Overhaul            ;Queue for service
          QUEUE         Alljobs             ;Collect global statistics
          SEIZE         Maintenance         ;Obtain TV repairman
          DEPART        Overhaul            ;Leave queue for man
          DEPART        Alljobs             ;Collect global statistics
          ADVANCE       600,60              ;Complete job 10+/-1 hours
          RELEASE       Maintenance         ;Free repairman
          TERMINATE     ;Remove one transaction
*        модельный сегмент 2
          GENERATE      90,10,,,3          ;On-the-spot repairs
```

```

        QUEUE      Spot                ;Queue for spot repairs
        QUEUE      Alljobs              ;Collect global statistics
        PREEMPT    Maintenance,PR      ;Get the TV repairman
        DEPART     Spot                ;Depart the 'spot' queue
        DEPART     Alljobs              ;Collect global statistics
        ADVANCE    15,5                ;Time for tuning/fuse/fault
        RETURN     Maintenance          ;Free maintenance man
        TERMINATE

*      модельный сегмент 3
        GENERATE   300,60,,,2          ;Normal TV Repairs
        QUEUE      Service              ;Queue for service
        QUEUE      Alljobs              ;Collect global statistics
        PREEMPT    Maintenance,PR      ;Preempt maintenance man
        DEPART     Service              ;Depart the 'service' queue
        DEPART     Alljobs              ;Collect global statistics
        ADVANCE    120,30               ;Normal service time
        RETURN     Maintenance          ;Release the man
        TERMINATE

*      модельный сегмент 4
        GENERATE   480                  ;One xact each 8 hr. day
        TERMINATE  1

*
Overhaul QTABLE   Overhaul,10,10,20
Spot     QTABLE   Spot,10,10,20
Service  QTABLE   Service,10,10,20
Alljobs  QTABLE   Alljobs,10,10,20

```

Задание 5

(Время выполнения — 50 мин)

Постановка задачи самостоятельной работы

Сделать модель работы мастерской по ремонту ноутбуков с двумя мастерами — системщиком и электроником. При диагностике системщиком отсеивается 7% посетителей как «неремонтируемые случаи». Основная часть ремонтов — исправление системщиком софтверных проблем (upgrade firmware, установка OS, анти-вирусы). Работы по ремонту электроники появляются в 30% случаев ремонта («сложный» ремонт) и их выполняют оба мастера совместно. Недостающие сведения можно взять из описания процесса в задании 4.

Необходимо промоделировать работу мастерской и построить график очереди работ в течение календарного месяца.

Лабораторная работа «Логические условия, семейства транзактов»
(Общее время выполнения — 4 часа)

Задание 1

(Время выполнения — 30 мин)

Изучите пример моделей SCHR7A1.GPS и SCHR7A2.GPS, в которых приведены разные варианты модели обслуживания клиентов в оперзале банка.

Вариант 1

```
Mean    FUNCTION    RN1,D5                ;Distribution of Mean Service Time
.1,450/.29,750/.61,1000/.85,1500/1,3000
Telrs    STORAGE    8                ;Provide 8 Tellers
*        модельный сегмент 1
        GENERATE    (Exponential(1,0,180)) ;Customers arrive
        ASSIGN      1,(Exponential(1,0,FN$Mean)) ;Set P1 = Service Time
        PRIORITY    1                ;Set high priority for later
        QUEUE       One              ;Enter the line
        GATE SNF     Telrs,Wait       ;Is a teller available
Grab      ENTER      Telrs            ;Yes capture a teller
        DEPART      One              ;Leave the line
        ADVANCE     P1               ;Transact business
        LEAVE       Telrs            ;Free the teller
        UNLINK      Line,Grab,1      ;Send next waiting customer
        TERMINATE    ;Leave the bank
Wait      LINK       Line,FIFO        ;No, go onto back of user chain
*        модельный сегмент 2
        GENERATE     180000          ;Timer arrives after 5 hours
        TERMINATE    1
```

Вариант 2

```
Mean    FUNCTION    RN1,D5                ;Distribution of Mean Service Time
.1,450/.29,750/.61,1000/.85,1500/1,3000
*        модельный сегмент 1
        GENERATE     (Exponential(1,0,180)) ;Customers arrive
        ASSIGN      1,(Exponential(1,0,FN$Mean)) ;Set P1 = service time
        PRIORITY    1                ;Set high priority for later
        SELECT E     2,1,8,0,F,Queuep    ;Is teller available?
Line      QUEUE      P2                ;Update segregated line data
        QUEUE      10                ;Update aggregate line data
        SEIZE       P2                ;Capture a teller
        DEPART      10                ;Update aggregate line data
        DEPART      P2                ;Update segregated line data
        ADVANCE     P1               ;Transact business
        RELEASE     P2               ;Free the teller
        TERMINATE    ;Leave the bank
Queuep    SELECT MIN 2,1,8,,Q          ;Set P2 = number of shortest line
        TRANSFER    ,Line            ;Go queue up
*        модельный сегмент 2
        GENERATE     180000          ;Timer arrives after 5 hours
        TERMINATE    1                ;Shut off the run
```

В варианте 1 очереди организованы отдельно к каждому операционисту, а в варианте 2 -очередь общая для всех. Необходимо открыть модели, получить результат моделирования за 10 рабочих смен. Оценить статистику по времени ожидания в очереди Line (среднее время и стандартное отклонение). Сравнить варианты ожидания в общей очереди и в очередях к каждому операционисту.

Получить сведения о загрузке операционистов. Разобрать алгоритм выборки значения в блоках SELECT.


Схема обслуживания

Время прихода -
Exponential (180)

5 типов операций :
10 % - 450 сек
19 % - 750 сек
32 % - 1000 сек
24 % - 1500 сек
15 % - 3000 сек

Время
обслуживания
индивидуально

операционисты



Задание 2

(Время выполнения — 40 мин)

Изучите пример модели FOUNDRY.GPS, в котором приведена модель литейной мастерской.

```

Weight      FUNCTION RN1,C8                                ;Weight per component in Kgs
0.0,3/.13,6/.25,11/.50,20/.70,28/.85,35/.95,42/1.0,50
Ordertype   FUNCTION RN1,D2                                ;New order P1=1: Repeat P1=2
0.3,1/1.0,2
Size        VARIABLE RN1@19+6                               ;Size of order
Ddate       VARIABLE V$Mtime#P2+RN1@121+40+C1 ;Due date
Mtime       VARIABLE (P3#2)                                ;Mold time per component
Day         VARIABLE (C1/480)                               ;Day indicator
Total       VARIABLE P3#P2                                  ;Weight per order
Times       TABLE M1,400,400,20                           ;Transit time
Cast        TABLE X$Wtmold,400,400,20                    ;Weight cast
Molders     STORAGE 18                                       ;Molders employed
*           модельный сегмент 1
            GENERATE (Exponential(1,0,60)) ;Jobs arrive every hour
            ASSIGN 1,FN$Ordertype ;Type of job
            TEST E P1,2,Newjob ;Is it a repeat order?
            ADVANCE 300,180 ;Locate pattern
Commence    ASSIGN 2,V$Size ;Size of order
            ASSIGN 3,FN$Weight ;Weight of component
            ASSIGN 4,V$Mtime ;Molding time per component
            ASSIGN 5,V$Ddate ;Due date
            ASSIGN 6,V$Total ;Total weight of order
            GATE SNF Molders,Wait ;Any molders free?
Beg         ENTER Molders ;Molder begins order
            ASSIGN 7,P2 ;P7=Number in order
Next        ADVANCE P4 ;Molding time per component
            LOOP 7,Next ;Loop for every component
            LEAVE Molders ;Free molder, order complete
            SAVEVALUE Wtmold+,P6 ;Sum weight molded each order
            UNLINK 1,Beg,1 ;Release next order
            TABULATE Times ;Tabulate transit time
    
```

	TERMINATE		;Destroy xact
Newjob	ADVANCE	4320,1440	;Time to make new pattern
	TRANSFER	,Commence	;Transfer to commence order
Wait	LINK	1,P5	;Link waiting orders in chain 1
*	модельный сегмент 2		
	GENERATE	420,,,1,2	;Start casting operation cycle
Again	SUNAVAIL	Molders	;Marks start of casting cycle
	ADVANCE	60	;Casting cycle lasts 60 mins
	SAVAIL	Molders	;Molders free for molding
	ADVANCE	420	;420 mins elapse before casting
	TABULATE	Cast	;Record total weight cast
	SAVEVALUE	Totcast+,X\$Wtmold	;Accumulate total cast so far
	SAVEVALUE	Wtmold,0	;Reset to zero each day
	TRANSFER	,Again	;Return xact to start again
*	модельный сегмент 3		
	GENERATE	4800,,,,,4	;Xact every ten days
	SAVEVALUE	V\$Day,X\$Totcast	;Records total weight cast
	TERMINATE	1	;Destroy xact

В литейной мастерской над мелкосерийными заказами, поступающими в среднем раз в час, работают 18 литейщиков. Мастерская работает в одну смену с 8-часовым рабочим днем. 30% заказов — новые, а 70% — повторные. Для новых заказов нужны новые формы и модели, которые делают в модельной мастерской за 72 ± 24 ч. Формы повторных заказов требуется найти и подготовить, что занимает 5 ± 3 ч. В заказе может быть от 6 до 24 штук деталей. Масса одной детали варьируется от 3 до 50 кг. Один заказ выполняет (формуется) один рабочий. Формовка занимает 2 мин на кг массы детали. Срок выполнения заказа определяется общим временем формовки плюс технологическое время от 40 до 160 ч на заказ. Заливка металла происходит раз в день в последний час смены. Заливку металла в заготовленные формы выполняют одновременно все рабочие.

Необходимо открыть модель, получить результат моделирования за 30 смен, собрать статистику по времени выполнения заказов (среднее время и станд. отклонение).

Оценить ежедневный расход металла (средняя масса и станд.отклонение), найти сведения о загрузке рабочих, построить графики очередей выполнения заказов по типам.

Задание 3

(Время выполнения — 40 мин)

Изучите пример модели ASSEMBLY.GPS, где приведена модель производственного участка сборки насосов.

Transit	TABLE	M1,200,200,20	
*	модельный сегмент 1		
	GENERATE	(Exponential(1,0,300))	;New order arrives
	SPLIT	2,Factory,1	;Make 2 copies of order
	QUEUE	Motor	;Queue for motor
	SEIZE	Motor	;Get a Facility
	DEPART	Motor	;Depart the queue
	ADVANCE	200,100	;Take motor from stock
	RELEASE	Motor	;Free the Facility

	TRANSFER	,Tryout	;Send to trial assembly
Factory	TEST E	P1,2,Baseplate	;Is P1=2 ?
	QUEUE	Pumps	;Join the Queue (P1=2)
	SEIZE	Pumps	;Get a Facility
	DEPART	Pumps	;Depart the Queue
	ADVANCE	180,120	;Prepare the Pump
Pump	MATCH	Plate	;Wait for baseplate
	ADVANCE	50,10	;Check pump on baseplate
	RELEASE	Pumps	;Free the Facility
	TRANSFER	,Tryout	;Send for a tryout
Baseplate	QUEUE	Base	;Join Queue P1 must=3
	SEIZE	Base	;Get a Facility
	DEPART	Base	;Depart the Queue
	ADVANCE	80,20	;Make the baseplate
Plate	MATCH	Pump	;Wait for the pump unit
	ADVANCE	50,10	;Check the pump on baseplate
	RELEASE	Base	;Free the Facility
Tryout	GATHER	3	;Gather 3 units to tryout
	ADVANCE	60	;Trial assembly
	TEST E	P1,1,Finish	;Is it the motor?(P1=1)
	SEIZE	Paint1	;Get first paint Facility
	ADVANCE	100,20	;Paint the motor
	RELEASE	Paint1	;Free paint Facility 1
	TRANSFER	,Build	;Send for assembly
Finish	TEST E	P1,2,Basplate	;Is it the pump?(P1=2)
	SEIZE	Paint2	;Get paint Facility 2
	ADVANCE	120,30	;Paint the Pump
	RELEASE	Paint2	;Free paint Facility 2
	TRANSFER	,Build	;Send for assembly
Basplate	SEIZE	Galvanize	;Get a Facility
	ADVANCE	120,30	;Galvanize baseplate
	RELEASE	Galvanize	;Free the Facility
Build	ASSEMBLE	3	;Collect 3 units
	ADVANCE	150,30	;Assemble unit
	TABULATE	Transit	;Record transit time
	TERMINATE	1	;One unit completed

На фабрике выпускают центробежные насосные агрегаты штучно по заказам клиентов. Заказы поступают в среднем каждые 5 часов. При получении заказа начинают 3 процесса: получения и подготовки двигателя, поиска и подгонки насоса, и изготовления станины. Когда станина готова, производится проверочный монтаж насоса. Затем делают пробную сборку агрегата из всех 3 компонентов. После проверочных работ, агрегат разбирается, насос и двигатель окрашиваются, а станина оцинковывается. В завершении все детали снова собираются.

Необходимо открыть модель; провести моделирование 100 заказов; исследовать коэффициенты использования сборочных участков; определить время выполнения заказов; построить график времени ожидания заказов до начала сборки.

Задание 4

(Время выполнения — 50 мин)

Изучите пример модели BICYCLE.GPS, где приведена модель работы велосипедной мастерской.

Orders	FUNCTION	P\$Department,L6	
1,Order/2,Frame/3,Saddle/4,Handlebars/5,Wheels/6,Pedals			
Transit	TABLE	M1,100,100,20	
Clerks	STORAGE	2	
Framers	STORAGE	3	
Saddlers	STORAGE	1	
Handlers	STORAGE	1	
Wheelers	STORAGE	1	
Pedalers	STORAGE	1	
Builders	STORAGE	4	
Packers	STORAGE	3	
* модельный сегмент 1			
	GENERATE	50,10	;Order arrives for bicycle
	SPLIT	5,Factory,Department	;Make 5 copies of order
Order	ENTER	Clerks	
	ADVANCE	80,10	;Prepare invoice
	LEAVE	Clerks	
Invoice	MATCH	Bicycle	;Synchronize with bicycle
	TERMINATE		;Transaction finished
* модельный сегмент 2			
Factory	TRANSFER	FN,Orders	;Route to correct dept.
* модельный сегмент 3			
Frame	ENTER	Framers	
	ADVANCE	(Exponential(1,0,65))	;Make frame
	ADVANCE	12,2	;Inspect frame
	LEAVE	Framers	
	TRANSFER	,Build	;Send for assembly
* модельный сегмент 3			
Saddle	ENTER	Saddlers	
	ADVANCE	6,3	;Get a saddle
	ADVANCE	3,1	;Inspect the saddle
	LEAVE	Saddlers	
	TRANSFER	,Build	;Send for assembly
* модельный сегмент 4			
Handlebars	ENTER	Handlers	
	ADVANCE	4,2	;Get handlebars
	ADVANCE	3,1	;Inspect handlebars
	LEAVE	Handlers	
	TRANSFER	,Build	;Send for assembly
* модельный сегмент 5			
Wheels	ENTER	Wheelers	
	ADVANCE	3,1	;Get wheels
	ADVANCE	3,1	;Inspect wheels
	LEAVE	Wheelers	
	TRANSFER	,Build	;Send for assembly
* модельный сегмент 6			
Pedals	ENTER	Pedalers	
	ADVANCE	5,1	;Get pedals
	ADVANCE	3,1	;Inspect pedals
	LEAVE	Pedalers	
	TRANSFER	,Build	;Send for assembly
* модельный сегмент 7			
Build	ASSEMBLE	5	;Assemble
	ENTER	Builders	
	ADVANCE	(Normal(1,90,10))	;Time for assembling
	ADVANCE	35,5	;Inspect
	LEAVE	Builders	
Bicycle	MATCH	Invoice	;Wait for paperwork
	ENTER	Packers	
	ADVANCE	40,5	;Pack for dispatch
	LEAVE	Packers	
	TABULATE	Transit	
	TERMINATE		;Transaction finished
* модельный сегмент 8			

GENERATE 480
TERMINATE 1

;Timer every day

В мастерской собирают велосипеды штучно по заказам клиентов. Заказы поступают в среднем каждые 50 минут. При получении заказа начинается 6 процессов по [заказаЗакаRO, [маб2 [юы алXСнаыньеш веЦ

Лабораторная работа «Проведение статистических экспериментов»

(Общее время выполнения — 4 часа)

Задание 1

(Время выполнения — 25 мин)

Изучите пример модели ANOVA.GPS, где на примере обслуживания клиентов парикмахером показан один из способов обработки результатов моделирования.

```
*      модельный сегмент 1
GENERATE 5,1.7                ;Create next customer.
QUEUE    Barber               ;Begin queue time.
SEIZE     Barber               ;Own or wait for barber.
DEPART    Barber               ;End queue time.
ADVANCE   Cut_Time             ;Cut takes a few min.
RELEASE   Barber               ;Give up the barber.
TEST G    TG1,1,Doout          ;Last XN writes Anova data.
TERMINATE 1                    ;Customer leaves.

*      модельный сегмент 2
Doout OPEN ("BARBER.RST"),,Prob ;Open Anova database.
SEEK      10000                ;Place pointer at end of file.
WRITE (Polycatenate(QT$Barber," ",Treatment," ")),,Prob,On
CLOSE     Errorcode,,Prob      ;Close,send to Prob if error.
TERMINATE 1

Prob TERMINATE 1                ;Terminate here if I/O error.
; INCLUDE "CTLANOVA.TXT"        ;Call runtime command file.
*****
```

Командный файл CTLANOVA.TXT

```
RESULTS MATRIX ,2,3 ; Set up for 3 replicates of two levels
Cut_Time EQU 6
Treatment EQU 1
RMULT 411
Start 100,NP
MSAVEVALUE RESULTS,1,1,QT$Barber
Clear Off
RMULT 421
Start 100,NP
MSAVEVALUE RESULTS,1,2,QT$Barber
Clear Off
RMULT 431
Start 100,NP
MSAVEVALUE RESULTS,1,3,QT$Barber
Clear Off
Cut_Time EQU 6.5
Treatment EQU 2
RMULT 411
Start 100,NP
MSAVEVALUE RESULTS,2,1,QT$Barber
Clear Off
RMULT 421
Start 100,NP
MSAVEVALUE RESULTS,2,2,QT$Barber
Clear Off
RMULT 431
Start 100,NP
MSAVEVALUE RESULTS,2,3,QT$Barber
```

Задача в примере — сравнить 2 варианта обслуживания клиентов в парикмахерской:

а) среднее время обслуживания = 6,8 мин

б) среднее время обслуживания = 5 мин

Критерием качества оценки является время ожидания клиентами начала обслуживания.

В постановке эксперимента на модели надо в качестве главного экспериментального фактора выбрать параметр Cut_Time с 2 уровнями изменения фактора. Количество прогонов (реplik) указать равным 3, и как исследуемый критерий изучить значение QT\$Barber.

Для выполнения эксперимента выполните команду «INCLUDE ctlanova.txt» через пункт меню «Command / Custom...».

Проверьте полученные результаты, записанные в матрицу Results через пункт меню «Window / Simulation window / Matrix window»

Вызовите библиотечную процедуру ANOVA через пункт меню «Command / SHOW... ANOVA(Results,2,1)».

Процедура ANOVA реализует метод дисперсионного анализа статистических результатов моделирования. Вызов процедуры с параметрами предусмотрен в виде *ANOVA (матрица, индекс для реплик, уровень взаимодействий факторов)*.

Задание 2

(Время выполнения — 25 мин)

Изучите пример модели ONEWAY.GPS, где на примере модели обслуживания звонков клиентов в телефонном центре проводится однофакторный дисперсионный анализ результатов моделирования.

```
Sets      STORAGE      2
Transit   TABLE      M1,.5,1,20      ;Transit times
*         модельный сегмент 1
          GENERATE     1.667,1        ;Calls arrive
Again     GATE SNF     Sets,Occupied   ;Try for a line
          ENTER        Sets           ;Connect call
          ADVANCE      3,1            ;Speak for 3+/-1 min
          LEAVE        Sets           ;Free a line
          TABULATE     Transit        ;Tabulate transit time
          TERMINATE    1              ;Remove a transaction
*         модельный сегмент 2
Occupied  ADVANCE      5,1            ;Wait 5 minutes
          TRANSFER     ,Again         ;Try again
```

Эксперимент разработан для изучения модели работы «горячей линии» с целью оценки влияния количества линий на качество обслуживания при помощи процедуры дисперсионного анализа ANOVA.

Экспериментальным фактором является размер storage Sets. Критерием качества является TB\$Transit. В эксперименте предполагается получение результатов моделирования для четырех уровней значения фактора — от 1 до 4 — с тремя репликами на каждом уровне.

В ходе проведения эксперимента результаты сохраняются в матрице результатов размерности 4 x 3 с именем MainResult. Для наблюдения за ходом выполнения прогонов можно через меню «View / Simulation Clock» включить отображение модельного времени.

Эксперимент запускается командой «CONDUCT BestLines(1,4,1)». После завершения выполнения прогонов можно проверить наличие результата через меню «Window / Simulation window / Matrix window». Помимо этого результат выгружался в файл «Result.TXT», который можно открыть через меню «File / Open... /» с типом файлов < *.txt>.

Дополнительно статистическое качество результата можно проверить, изучив состояние таблицы MainResult_Residuals через меню «Window / Simulation window / Table window».

Задание 3

(Время выполнения — 15 мин)

Изучите пример модели MULTIWAY.GPS, где приведен пример запуска многофакторного дисперсионного анализа.

Этот эксперимент разработан для заполнения матрицы результатов и ее дисперсионного анализа библиотечной процедурой ANOVA.

Для запуска примера нажмите клавишу F11, выполняющую команду «SHOW (FillMatrix())».

Затем нажмите клавишу F12, за которой закреплена команда «SHOW (ANOVA(MainResult, 3, 2))»

Для изучения матрицы MainResult воспользуйтесь пунктом меню «Window / Simulation window / Matrix window» и откройте 2 разных сечения матрицы MainResult.

Дополнительно статистическое качество результата можно проверить, изучив состояние таблицы MainResult_Residuals через меню «Window / Simulation window / Table window».

Задание 4

(Время выполнения — 15 мин)

Изучите пример модели LATINSQUARE.GPS, где приведен пример организации многофакторного дисперсионного анализа.

Этот эксперимент разработан для демонстрации трехфакторного эксперимента и статистического анализа матрицы результатов без учета влияния взаимодействий факторов.

Для запуска примера нажмите клавишу F11, выполняющую команду «SHOW (FillMatrix())».

Затем нажмите клавишу F12, за которой закреплена команда «SHOW (ANOVA(MainResult, 0, 1))»

Для изучения матрицы MainResult воспользуйтесь пунктом меню «Window / Simulation window / Matrix window» и откройте 3 разных сечения матрицы MainResult.

Дополнительно статистическое качество результата можно проверить, изучив состояние таблицы MainResult_Residuals через меню «Window / Simulation window / Table window».

Задание 5

(Время выполнения — 30 мин)

Изучите пример модели EXPERETHER.GPS, где на примере модели функционирования сегмента EtherNet показана подготовка отсеивающего эксперимента.

```
Node_Count      EQU 100      ;Total Ethernet Nodes
Intermessage_Time EQU 1.0    ;Avg. Global Arrival every msec.
Min_Msg         EQU 512      ;The Shortest Message in bits
Max_Msg         EQU 12144    ;The Longest Message in bits
Fraction_Short_Msgs EQU 600  ;Short Msgs in parts per thousand
Slot_Time       EQU 0.0512   ;512 bit times
Jam_Time        EQU 0.0032   ;32 bit times
Backoff_Limit   EQU 10       ;No more than 10 backoffs
Interframe_Time EQU 0.0096   ;96 bit times
Backoff_Delay   VARIABLE Slot_Time#V$Backrandom
Backrandom      VARIABLE 1+(RN4@((2^V$Backmin)-1))
Backmin         VARIABLE (10#(10'L'P$Retries))+(P$Retries#(10'GE'P$Retries))
Node_Select     VARIABLE 1+(RN3@Node_Count)
Collide         VARIABLE ABS((X$Xmit_Node-P$Node_ID)/100000)'GE'(AC1-X$Xmit_Begin)
Msgtime        VARIABLE (0.0001)#V$Msgrand
Msgrand         VARIABLE Min_Msg+(RN1'G'Fraction_Short_Msgs)#(Max_Msg-Min_Msg)
Msg_Delays      QTABLE Global_Delays,1,1,20
```

```
*      модельный сегмент 1
GENERATE (Exponential(1,0,Intermessage_Time)) ;Single generator
ASSIGN   Node_ID,V$Node_Select ;Acquire a Node ID.
ASSIGN   Message_Time,V$Msgtime ;Calc and Save XMIT Time.
ASSIGN   Retries,0 ;No Collisions at start.
QUEUE    Global_Delays ;Start timing
SEIZE    P$Node_ID ;Wait for, occupy, the Node.
Try_To_Send PRIORITY 1 ;Don't Lose Control
SEIZE    Jam ;Wait for any
RELEASE  Jam ;Jam to end.
TEST E   F$Ethernet,1,Start_Xmit ;If Ethernet Free, jump.
TEST E   V$Collide,1,Start_Xmit ;No Collision. Go Wait for it.

*      модельный сегмент 2
Collision PREEMPT Ethernet,PR,Backoff,,RE ;Remove the old owner.
SEIZE    Jam ;Jam the Ethernet.
ADVANCE  Jam_Time ;Wait the Jam Time.
RELEASE  Jam ;End the Jam.
RELEASE  Ethernet ;Give up the Ethernet.
PRIORITY 0 ;Back to Normal priority.
Backoff  ASSIGN Retries+,1 ;Increment the Backoff Ct.
TEST LE  P$Retries,Backoff_Limit,Xmit_Error ;Limit retries.
ADVANCE  V$Backoff_Delay ;Wait to initiate retry.
TRANSFER ,Try_To_Send ;Go try again.

*      модельный сегмент 3
Start_Xmit SEIZE Ethernet ;Get Ethernet, wait if nec.
SAVEVALUE Xmit_Node,P$Node_ID ;Identify the sender.
```

```

SAVEVALUE Xmit_Begin,AC1      ;Mark the start xmit time.
PRIORITY 0                    ;Ensure we can be PREEMPTed.
ADVANCE P$Message_Time        ;Wait until Msg. is sent.
ADVANCE Interframe_Time       ;Hold the Ethernet for gap.
RELEASE Ethernet              ;Give up the Ethernet.
Free_Node RELEASE P$Node_ID    ;Give up the node
DEPART Global_Delays          ; to the next msg.
TERMINATE                     ;Destroy the Message.
Xmit_Error SAVEVALUE Error_Count+,1 ;Count the Error.
TRANSFER ,Free_Node          ; and get out of the way.
*      модельный сегмент 4
      GENERATE 1000            ;Each Start Unit is 1 Second.
      TERMINATE 1

```

Этот эксперимент разработан для определения наиболее значимых параметров в модели.

В качестве тестовой модели служит модель сегмента «тонкого» Ethernet.

Задача эксперимента сводится к поиску факторов, наиболее влияющих на суммарную производительность сети. В качестве критерия качества выбрано время передачи пакета между узлами сети.

В постановке эксперимента на модели заявлены такие экспериментальные факторы — Intermessage_Time, Node_Count, Min_Msg, Max_Msg, Fraction_Msgs.

В качестве исследуемого критерия объявлен параметр QT\$Global_Delays. Количество прогонов принято равным 16 (т.е. половина полного эксперимента).

Для проведения эксперимента вызовите пункт меню «Edit / Insert Experiment / Screening...». Настройте параметры эксперимента в соответствии с вышеизложенной постановкой.

Для запуска эксперимента выполните команду «CONDUCT ScreenEthernet()», продублированную на клавише F11.

После завершения выполнения прогонов результат будет виден в окне журнала моделирования. Дополнительно можно изучить результаты по матрице результатов ScreenEthernet_Results размерностью (2 x 2 x 2 x 2 x 2).

Для просмотра матриц воспользуйтесь пунктом меню «Window / Simulation window / Matrix window», а для проверки значений прочих параметров обратитесь к пункту меню «Window / Simulation window / Table window».

Задание 6

(Время выполнения — 30 мин)

Изучите пример модели EXPERETHER.GPS, где на примере модели функционирования сегмента EtherNet показана подготовка оптимизирующего эксперимента.

Этот эксперимент разработан для поиска аппроксимирующей поверхности отклика методом наискорейшего подъема. Подобранные уравнение поверхности отклика является необходимым уравнением для построения прогнозов. Помимо этого, результаты могут использоваться и для выбора оптимальных значений параметров модели.

В качестве тестовой модели служит модель сегмента «тонкого» Ethernet. Задачей является поиск уравнения поверхности отклика в некоторой локальной области существования результата. Критерием качества выбрано время передачи пакета.

В постановке этого эксперимента на модели экспериментальными факторами являются *Intermessage_Time*, *Node_Count*, *Max_Msg*, *Fraction_Short_Msgs*.

В качестве исследуемого критерия рекомендуется выбрать величину $(1/QT\$Global_Delays)$, чтобы тип оптимального экстремума был поиск экстремума типа максимум.

Для подготовки эксперимента воспользуйтесь пунктом меню «Edit / Insert Experiment / Optimizing...». Обратите внимание на настройку в процедуре *DoTheRun* счетчиков прогонов.

Эксперимент запускается командой «CONDUCT RSM()» или нажатием клавиши F12.

После завершения выполнения прогонов результат будет виден в окне журнала моделирования. Дополнительно можно изучить результаты по матрицам результатов – *RSM_BestYields*, *RSM_CurrentBestTC*, *RSM_OptimumTC*. Для просмотра матриц воспользуйтесь пунктом меню «Window / Simulation window / Matrix window».

Задание 7

(Время выполнения — 30 мин)

Подготовьте проведение оптимизирующего эксперимента на основе следующей модели.

```
NWorkers EQU 2 ; количество рабочих
Machine Storage 1
* модельный сегмент 1
GENERATE ,,,Nworkers ; транзакты -это рабочие
Back1 ADVANCE 20,7 ; рабочий собирает изделие
ENTER Machine ; используется станок
ADVANCE 15,3 ; время работы станка
LEAVE Machine ; освобождается станок
Out TRANSFER ,Back1 ; новая сборка изделия
* модельный сегмент 2
GENERATE (60#8#5) ; 60 мин x 8 часов x 5 дней
TERMINATE 1
```

Проведите эксперимент для определения, сколько нужно рабочих и станков для максимума итогового дохода. В качестве факторов эксперимента используйте параметр *NWorkers* в диапазоне от 1 до 11, и параметр *mach* в диапазоне от 1 до 3. Критерием оптимизации будет являться результат выражения формулы дохода $(N\$Out\#50-NWorkers\#2000-mach\#700)$.

Для связи параметров *mach* и *Machine* добавьте в процедуру *DoTheRun* команду вида

```
DoCommand (catenate («Machine Storage », mach)).
```

Приложение 1. Структура оформления отчета по лабораторной работе

ФИО студента _____ группа _____ дата _____

ЛАБОРАТОРНАЯ РАБОТА

« _____ т_е_м_а _____ р_а_б_о_т_ы _____ »

Название модели: _____

Структура модели:

< схема в виде системы массового обслуживания >

< т.е. граф связей : источники — буферы — обслуживающие аппараты — терминаторы >

Текст стандартного отчета о результатах прогона:

< текст из Gpss World >

Графические результаты выполнения модели:

< диаграммы, графики, таблицы из Gpss >

Вывод по результатам использования модели:

< ответы на поставленные в задании вопросы; какие характеристики можно рассчитать; какие зависимости можно анализировать; какие состояния можно визуализировать; какие аналитические методы возможно использовать для этой модели >